#include "include\\Common.fxh"
/*
% Static Evironment Mapping.
date: 06/19/12
*/

/*****************Un-Twealables**********************************/
#define FLIP_TEXTURE_Y 1

//after colon is a semantic usefull later for knowing what variables this shader has
float4x4 WorldViewProjection : WORLDVIEWPROJECTION < string UIWidget="None";>;
float4x4 WorldInverseTranspose : WORLDINVERSETRANSPOSE < string UIWidget="None";>;
float4x4 World:WORLD <string UIWidget = "None";>;
float4x4 ViewInverse:VIEWINVERSE <string UIWidget = "None";>;

/****************Tweakables************************************/
texture ColorTexture : COLOR 
< 
	string ResouceName = "default_color.dds";
	string UIName = "Color Texture";
	string ResourceType = "2D";
>;

float4 AmbientColor : COLOR
<
	string UIName = "Ambient Color";
	string UIWidget = "Color";
	
> = {1.0f,1.0f,1.0f,1.0f}; //default value

float4 EnvColor
<
	string UIName = "Environment Color";
	string UIWidget = "Color";
> = {1.0f,1.0f,1.0f,1.0f}; //default value
/***************Un-Tweakable************************************/
sampler2D ColorSampler = sampler_state
{	
	Texture = <ColorTexture>; //the texture it will be grabbing pixel date from
	Magfilter = LINEAR;		  //zoom in - does linear interpelation
	Minfilter = LINEAR;		  //zoom out
	Mipfilter = LINEAR;
	AddressU = Wrap;		  //wrap or clamp
	AddressV = Wrap;		  //uv values or 0 to 1 oustide that they must either be wrapped(round up to the next) or clamped(made to fit in 0to1)
};

texture EnvironmentMap : ENVMAP <
	string UIName = "Environment Map";
	string ResourseType = "3D";
>;

sampler EnvironmentMapSampler = sampler_state
{
	Texture = <EnvironmentMap>;
	Magfilter = LINEAR;		  //zoom in - does linear interpelation
	Minfilter = LINEAR;		  //zoom out
	AddressU = Clamp;		  //wrap or clamp
	AddressV = Clamp;		  //uv values or 0 to 1 oustide that they must either be wrapped(round up to the next) or clamped(made to fit in 0to1)
	AddressW = Clamp;
};

/***********************FOG**********************************/
float3 FogColor
< string UIName = "Fog Color";
  string UIWidget = "Color";
> = {0.5f,0.5f,0.5f};
float FogStart = {20.0f};
float FogEnd = {40.0f};
bool FogEnabled = true;
/*****************Data Structures****************************/
struct VS_INPUT
{
	float4 ObjectPosition: POSITION0;
	float3 Normal : NORMAL0;
	float2 TextureCoordinate : TEXCOORD0;
};
struct VS_OUTPUT
{
	float4 ScreenPosition: POSITION0;
	float2 TextureCoordinate: TEXCOORD0;
	float3 ReflectionVector: TEXCOORD1;
	float  FogAmount : TEXCOORD3;
};
struct PS_OUTPUT
{
	float4 Color:COLOR0;
};

/*****************Vertex Shader*****************************/
VS_OUTPUT vertex_shader(VS_INPUT IN)
{
	VS_OUTPUT OUT = (VS_OUTPUT)0; //memset clear garbage
	OUT.ScreenPosition = mul(IN.ObjectPosition, WorldViewProjection);
	OUT.TextureCoordinate = get_corrected_texture_coordinate(IN.TextureCoordinate);
	
	float4 worldPosition = mul(IN.ObjectPosition,World);
	float3 eyePosition = ViewInverse[3].xyz;
	float3 incident  = normalize(worldPosition - eyePosition);
	float3 normal = normalize(mul(IN.Normal, World));
	
	//Relfection Vector for cube map: R = I - 2*N * (I.N)
	OUT.ReflectionVector = reflect(incident,normal);
	
	OUT.FogAmount = get_fog_amount(eyePosition, worldPosition, FogStart, FogEnd);
	
	return OUT;
}

/*****************Pixel Shader*******************************/
PS_OUTPUT pixel_shader(VS_OUTPUT IN)
{
	PS_OUTPUT OUT = (PS_OUTPUT)0;
	float4 color = tex2D(ColorSampler, IN.TextureCoordinate);
	float4 ambient = get_color_contribution(AmbientColor,color);
	float4 reflection = get_color_contribution(EnvColor, texCUBE(EnvironmentMapSampler, IN.ReflectionVector));
	
	OUT.Color = ambient + reflection;
	if(FogEnabled)
	{
		OUT.Color.rgb = apply_fog(FogColor, OUT.Color, IN.FogAmount);
	}
	return OUT;
}

/****************Techniques**********************************/
technique main
{
	pass p0 
	{
		VertexShader = compile vs_3_0 vertex_shader();
		PixelShader = compile ps_3_0 pixel_shader();
	}
}
